
Subject: Re: days of the week

Posted by [David Fanning](#) on Wed, 20 Mar 2002 15:47:12 GMT

[View Forum Message](#) <> [Reply to Message](#)

Bastienne Schneiter (b.schneiter@meteonews.ch) writes:

> I need to show a date in a little application. Its format is
> YYYYMMDDHH. But I also have to show to which day of the week this date
> correspond (MON, TUE etc).
> Does anybody know how to determine the day of the week for a given
> date?

Alright. Here is a little function that, given the
julian date, will tell you what day of the week it
is. For example, suppose you want to get married
on November 17, 2002 and the preacher asks you what
day of the week that is. You do this:

"Uh, hold-on...."

```
IDL> juliandate = Julday(11, 17, 2002)
IDL> Print, WhatDayIsIt(juliandate)
      Sunday
```

"Uh, that's a Sunday, Man."

Cheers,

David

P.S. Sorry for the FOR loop. I was in a hurry. :-(

```
FUNCTION WhatDayIsIt, juliandate
```

```
    ; Need a date? Duh...
```

```
IF N_Elements(juliandate) EQ 0 THEN $
    juliandate = Systime(/Julian)
```

```
    ; Make a table. Use week of March 17th, 2002.
```

```
daysOfWeek = ['Sunday', 'Monday', 'Tuesday', 'Wednesday', $
               'Thursday', 'Friday', 'Saturday']
datemod = IntArr(7)
```

```
FOR j=0,6 DO BEGIN
    jdate = Julday(3, 17 + j, 2002)
```

```
    datemod[j] = jdate MOD 7  
ENDFOR
```

```
    ; Convert to day, month, year.
```

```
CalDat, juliandate, month, day, year  
jdate = Julday(month, day, year)
```

```
    ; What day of the week is it? Return it.
```

```
index = Where(datemod EQ (jdate MOD 7))  
RETURN, daysOfWeek[index]  
END
```

```
*****
```

```
--
```

```
David W. Fanning, Ph.D.  
Fanning Software Consulting  
Phone: 970-221-0438, E-mail: david@dfanning.com  
Coyote's Guide to IDL Programming: http://www.dfanning.com/  
Toll-Free IDL Book Orders: 1-888-461-0155
```
